
INTRODUCTION 
This paper focuses on using SAS to facilitate two important project management techniques – evaluation and forecasting.  The 
objective of these techniques is to explain and predict project duration and resource consumption at a specific point in time.  The benefit 
of these techniques is that you can determine whether your initial work and spending schedule is on target well before the project ends. 
 
SCENARIO 
You are ten months into a project.  On the first day of each month, a new task was scheduled to start.  Each task was budgeted a 
specific amount of money and was had an estimate of how much should have been completed at the end of the month.  The actual 
amount spent and the actual amount completed was also tracked; the information tracked was as follows: 
 

Performance 
Period 

Scheduled 
Task 

Task Cost 
Estimates 

Actual Costs Planned % 
Complete 

Actual % 
Complete 

January A 1000 1100 1.0 1.0 
February B 2000 1750 .9 .95 

March C 3000 2320 .8 .83 
April D 4000 2750 .7 .65 
May E 5000 3150 .6 .55 
June F 6000 2950 .5 .53 
July G 7000 2615 .4 .47 

August H 8000 2502 .3 .22 
September I 9000 1703 .2 .15 

October J 10000 730 .1 .05 
 
Using these inputs, you calculate the Planned Value (PV) and the Earned Value (EV).  PV represents the amount of value expected to 
have been realized at a specific period.  EV represents the amount of actual value realized at a specific period.  PV and EV are derived 
by multiplying the task cost estimate by the planned and actual completion percentages respectfully.  When you calculate the PV and 
EV and sum them, the task cost estimates and the actual costs, you then have all of the inputs needed for evaluation and forecasting: 
 

Performance 
Period 

Scheduled 
Task 

Task Cost 
Estimates 

Actual 
Costs 

Planned % 
Complete 

Actual % 
Complete 

Planned 
Value 

Earned 
value 

January A 1000 1100 1.0 1.0 1000 1000 
February B 2000 1750 .9 .95 1800 1900 

March C 3000 2320 .8 .83 2400 2490 
April D 4000 2750 .7 .65 2800 2600 
May E 5000 3150 .6 .55 3000 2750 
June F 6000 2950 .5 .53 3000 3180 
July G 7000 2615 .4 .47 2800 3290 

August H 8000 2502 .3 .22 2400 1760 
September I 9000 1703 .2 .15 1800 1350 

October J 10000 730 .1 .05 1000 500 
TOTAL  55000 21570   22000 20820 

 
The code segment below mimics this scenario and provides the following evaluation and forecasting information in an email project 
status report: 
 
EVALUATION – How Well Is the Project Performing Now? 
 

• BAC (Budget At Completion) – The original cumulative cost for the project; the same as the summed task cost estimates. 
• CV (Cost Variance) – The difference between the estimated values of the work completed (EV) and what is actually paid out 

for the work (AC). 
• SV (Schedule Variance) – The difference between the estimated value of the work actually completed (EV) and the planned 

value completed (PV).  The SV is expressed in dollars.  Using earned value analysis, you can measure schedule progress in 
terms of dollars because the completed work is defined by a dollar amount.  Hence, this is the spending schedule or amount of 
time that a project is ahead or behind. 

• PC (Percent Complete) – The percentage of work completed to date compared with the planned work for the entire project. 
• PS (Percent Spent) – The percentage of money spent to date compared with the budget for the entire project. 
• CPI (Cost Performance Index) – The value of work completed for each dollar actually spent. 

• SPI (Schedule Performance Index) – The value of work completed for each dollar planned to be spent. 
 
FORECASTING – How Well Will the Project Perform? 
 

• EAC (Estimate At Completion) – The new/adjusted cumulative estimate of how much the project will cost. 
• VAC (Variance At Completion) – The difference between the original estimate and the new/adjusted estimate of the cumulative 

completion cost. 
• ETC (Estimate To Complete) – The estimate of how much money will be needed to operate the project from today until the 

end of the project. 
• TCPI_B (To Complete Performance Index – BAC) – An index of how much effort will be needed to get the project back on 

track based on the original budget (BAC).  It represents the relationship between the remaining work and the remaining money 
to complete the work. 

• TCPI_E (To Complete Performance Index – EAC) – An index of how much effort will be needed to get the project back on 
track based on the new/adjusted budget (BAC).  It represents the relationship between the remaining work and the remaining 
money to complete the work, assuming that the original budget has been re-baselined to the current budget. 

 
For this example, the project status information is stored in an ExcelP

®
P worksheet.  SAS will read this file to capture the inputs: 

CODING APPROACH 
%macro eval_fore;                                                                                                       
  proc datasets kill;            /*delete pre-existing temporary data sets*/                                            
  run;                                                                                                                  
  quit;                                                                                                                 
                                                                                                                        
  %macro calculate(period=,      /*performance period*/                                                                 
                   task_name=,   /*task id*/                                                                            
                   cost_est=,    /*original budget*/                                                                    
                   act_cost=,    /*actual cost*/                                                                        
                   plan_comp=,   /*planned completion percentage*/                                                      
                   act_comp=);   /*actual completion percentage*/                                                       
                                                                                                                        
    data one;                                                                                                           
      period="&period";                                                                                                 
      task="&task_name";                                                                                                
      cost_est=&cost_est;                                                                                               
      act_cost=&act_cost;                                                                                               
      plan_comp=&plan_comp;                                                                                             
      act_comp=&act_comp;                                                                                               
      pv=round(cost_est*plan_comp); /*planned value*/                                                                   
      ev=round(cost_est*act_comp);  /*earned value*/                                                                    
    run;                                                                                                                
                                                                                                                        
    proc append base=all_tasks data=one force;                                                                          
    run;                                                                                                                
                                                                                                                        
  %mend calculate;                                                                                                      
                                                                                                                        
  %macro importit;  /*Import the input values from an EXCEL file */                                                     
                                                                                                                        
    %macro obsnvars(ds);  /*metadata macro that captures data set information*/                                         
      %global dset nvars nobs;                                                                                          
      %let dset  = &ds;                                                                                                 
      %let dsid  = %sysfunc(open(&dset));                                                                               
      %let nobs  = %sysfunc(attrn(&dsid,NOBS));                                                                         
      %let nvars = %sysfunc(attrn(&dsid,NVARS));                                                                        
      %let rc    = %sysfunc(close(&dsid));                                                                              
    %mend obsnvars;                                                                                                     
                                                                                                                        
    proc import datafile="c:\bkb\test\test.xls"                                                                         
      out=test dbms=excel replace;                                                                                      
    run;                                                                                                                
    quit;                                                                                                               
                                                                                                                        
    %obsnvars(test);                                                                                                    
                                                                                                                        
    %do n = 1 %to &nobs;  /*for each observation, calculate the evaluation and forecasting inputs */                    
      data _null_;                                                                                                      
        set test;                                                                                                       
        if _N_ = &n;                                                                                                    
        call symput('perd',period);                                                                                     
        call symput('task',task_name);                                                                                  
        call symput('estm',cost_est);                                                                                   
        call symput('act1',act_cost);                                                                                   
        call symput('plan',plan_comp);                                                                                  
        call symput('act2',act_comp);                                                                                   
      run;                                                                                                              
                                                                                                                        
      
%calculate(period=&perd.,task_name=&task.,cost_est=&estm.,act_cost=&act1.,plan_comp=&plan.,act_comp=&act2.);            
                                                                                                                        
    %end;                                                                                                               
  %mend importit;                                                                                                       
  %importit;                                                                                                            
                                                                                                                        
  proc means data=all_tasks sum noprint;                                                                                
    output out=two(keep=cost_est act_cost pv ev rename=(cost_est=bac)) sum=;                                            
  run;                                                                                                                  
                                                                                                                        
  data three;                                                                                                           
    set two;                                                                                                            
    cv=ev-act_cost;                      /*cost variance*/                                                              
    sv=ev-pv;                            /*schedule variance*/                                                          
    cpi=round((ev/act_cost),.01);        /*cost performance index*/                                                     
    spi=round((ev/pv),.01);              /*schedule performance index*/                                                 
    pc=round((ev/bac),.01);              /*percent complete*/                                                           

    ps=round((act_cost/bac),.01);        /*percent spent*/                                                              
    eac=round((bac/cpi),.01);            /*estimate at completion*/                                                     
    vac=round(bac-eac);                  /*variance at completion*/                                                     
    etc=round(eac-act_cost);             /*estimate to complete*/                                                       
    tcpi_b=round(((bac-ev)/(bac-act_cost)),.01); /*to complete performance index - bac*/                                
    tcpi_e=round(((bac-ev)/(eac-act_cost)),.01); /*to complete performance index - eac*/                                
  run;                                                                                                                  
                                                                                                                        
  data _null_;                            /*macrotize the outputs*/                                                     
    set three;                                                                                                          
    call symput('bac',bac);                                                                                             
    call symput('act_cost',act_cost);                                                                                   
    call symput('ev',ev);                                                                                               
    call symput('pv',pv);                                                                                               
    call symput('cv',cv);                                                                                               
    call symput('sv',sv);                                                                                               
    call symput ('cpi',cpi);                                                                                            
    call symput ('spi',spi);                                                                                            
    call symput ('pc',pc);                                                                                              
    call symput ('ps',ps);                                                                                              
    call symput ('eac',eac);                                                                                            
    call symput ('vac',vac);                                                                                            
    call symput ('etc',etc);                                                                                            
    call symput ('tcpi_b',tcpi_b);                                                                                      
    call symput ('tcpi_e',tcpi_e);                                                                                      
  run;                                                                                                                  
                                                                                                                        
  data _null_;                         /*send results in email */                                                       
    filename outbox email "beverly_b@bls.gov";                                                                          
    file outbox                                                                                                         
    subject="Project Status Report";                                                                                    
    put "This is the status of your project, as of &sysdate :";                                                         
    put "                                                    ";                                                         
    put "The original estimated budget needed to complete the project was %cmpres(&bac).";                              
    put "The actual amount spent is %cmpres(&act_cost).";                                                               
    put "You planned to have earned %cmpres(&pv) worth of value, but the actual earned value is %cmpres(&ev).";         
    put "Based upon the earned value, the cost variance is %cmpres(&cv) (+ = under budget/- = over budget) and";        
    put "the schedule variance is %cmpres(&sv) (+ = ahead of spending schedule/- = behind spending schedule).";          
    put "The project is %cmpres(&pc) complete and %cmpres(&ps) of the budget has been spent.";                          
    put "The Cost Performance Index is %cmpres(&cpi) - this represents the ROI for every dollar spent.";                
    put "The Schedule Performance Index is %cmpres(&spi) - this represents the schedule performance against the 
plan.";                                                                                                                 
    put "Based upon the current performance, this project will cost %cmpres(&eac) to complete.";                        
    put "The difference between the original budget estimate and the current budget estimate is %cmpres(&vac).";        
    put "You will need %cmpres(&etc) (+ = more/ - = less) to complete this project.";                                   
    put "Based upon the original budget, the To Complete Performance Index (work remaining/money remaining) is 
%cmpres(&tcpi_b) .";                                                                                                    
    put "If you re-baseline the project current budget, then the ";                                                     
    put "To Complete Performance Index (work remaining/money remaining) would be %cmpres(&tcpi_e) .";                   
    put "                                                    ";                                                         
    put "Thanks - The SAS Project Management Office";                                                                   
  run;                                                                                                                  
                                                                                                                        
%mend eval_fore;                                                                                                        
%eval_fore; 
 
 
The email report is as follows: 

These results suggest that the project is slightly over budget and slightly behind schedule at the time of this report.  However, by adding 
$1,701 to the total budget, there will be enough money to cover the remaining work.  Another option would be to maintain the original 
budget, and improve the performance of the under-performing tasks.  For example, if you can redesign work processes, motivate 
project staff and reduce waste for Tasks H, I and J, such that the actual completion percentages are 27%, 18% and 8% respectively, 
then outcomes are dramatically different: 
 

 
 
Hence, by slightly improving the performance of three tasks, the whole project moved within $545 (1%) of being on target of the 
baseline estimate ($55,000).  As a follow up, it would be worthwhile to target Tasks D, E, H, I and J for additional analysis.  Determining 

why these tasks under-performed (as per their respective Earned Value schedules) would serve as the agenda for a risk/scheduling 
meeting. 
 
CONCLUSION 
The main take-away from this paper is that you can combine SAS with project management tools to proactively improve project and 
task performance.  Specifically, by codifying industry-standard project management techniques, managers and technologists can work 
together in monitoring the schedule and cost of producing SAS-based deliverables.  The benefit of using these duration and resource 
surveillance tools is that challenges to success can be addressed while they are still risks and before they become problems.  

